



SASで始めよう

被験者数再計算を伴う生物学的同等性試験

～ 後発医薬品のBEガイドライン改正に適應する ～

森田 祐介, 浜田 泉

(ノーベルファーマ株式会社 データサイエンス室)

Let's Try Two-stage Bioequivalence Trials with
Sample Size Re-estimation using SAS

Yusuke Morita, Izumi Hamada
Data Science Office, Nobelpharma Co., Ltd.

要旨：

令和2年3月，後発医薬品の生物学的同等性試験ガイドラインの改正により
中間解析結果に基づく被験者数追加が1回に限り認められることとなった
本発表では被験者数再計算を伴う2段階デザインに対応した解析手法として
Maximum Combination Testおよび被験者数再計算アルゴリズムを
関連手法とともに紹介し，SASによる実装方法を提案する

キーワード：生物学的同等性試験，2段階デザイン，アダプティブデザイン，第一種の過誤，
逆正規法，被験者数再計算，Maximum Combination Test，Potvin，中間解析，
後発医薬品の生物学的同等性試験ガイドライン





発表内容

1. 背景
2. BEガイドラインの改正に対応しうる解析手法
3. SASによる実装方法の提案
4. その他，考慮すべき事項

生物学的同等性試験とは

有効成分が同じ医薬品について
臨床上の有効性および安全性が同等であるか評価



後発医薬品のBEガイドラインの改正

より厳密な第一種の過誤確率の制御が必要に

改正前

改正後^[1,2]
(令和2年3月)

予試験

評価資料として使用可

評価資料として使用不可

本試験

同等でない場合に
例数追加試験を実施可

第一種の過誤確率を制御したうえで
中間解析や被験者数の追加可

例数追加試験

本試験の半分以上の
例数で実施可

原則不可

BE試験での第一種の過誤とは


- 本当は生物学的に同等でない製剤同士を試験結果から誤って同等であると判定すること
- BEガイドラインは、2つの片側検定で第一種の過誤確率を各々5%以下に維持することを要求
 - 両側90%信頼区間による評価と同等

本発表で想定する試験デザイン

中間解析を含む2段階デザイン
2剤2期のクロスオーバー試験

同等でない場合

被験者数を再計算して継続 又は 無効中止



本発表で用いる表記 (I)

- BE試験における帰無仮説と対立仮説

(複合)帰無仮説 : $H_{01} : \delta \leq \log(0.80) \cup H_{02} : \delta \geq \log(1.25)$

(複合)対立仮説 : $H_{11} : \delta > \log(0.80) \cap H_{12} : \delta < \log(1.25)$

対数変換した同等性評価パラメータの

試験製剤の平均を μ_T , 対照製剤の平均を μ_R

$\delta = \mu_T - \mu_R$ とする

説明の便宜上

ポイント帰無仮説 : $H_{01}^* : \delta = \log(0.80)$, $H_{02}^* : \delta = \log(1.25)$

と定義

本発表で用いる表記 (2)

- 帰無仮説は, H_{01} vs H_{11} および H_{02} vs H_{12} に対応する
2つの片側検定(TOST)によって検定

H_{01} と H_{02} に対するステージ i の検定統計量は以下となる

$$T_{i1} = \frac{\log(1.25) + \hat{\delta}_i}{\sqrt{2\hat{\sigma}_i^2/n_i}}, \quad T_{i2} = \frac{\log(1.25) - \hat{\delta}_i}{\sqrt{2\hat{\sigma}_i^2/n_i}}$$

$\hat{\delta}_i$: ステージ i の $\delta = \mu_T - \mu_R$ の推定値

$\hat{\sigma}_i^2$: ステージ i の対数変換した同等性評価パラメータの被験者内分散 σ^2 の推定値

n_i : ステージ i の被験者数

T_{i1} と T_{i2} はそれぞれ自由度 $n_i - 2$ の t 分布に従い,
対応する p 値を p_{i1} と p_{i2} とする

2段階デザインの統計的課題

- 2つの要因により第一種の過誤確率が上昇^[11]
 1. 中間解析と最終解析の繰り返しによる上昇
 2. 被験者数再計算による上昇
- 第一種の過誤確率を制御可能な解析手法が必要




発表内容

1. 背景
2. BEガイドラインの改正に対応しうる解析手法
3. SASによる実装方法の提案
4. その他，考慮すべき事項

2段階デザインに対応する解析手法

- Potvinの方法^[8]
- Standard Combination Test^[15]
- Maximum Combination Test^[10]
- 群逐次法(被験者数再計算を伴わないため対象外)
- 被験者数再計算アルゴリズム^[10]

Potvinの方法 (Method B)



- 中間解析と最終解析で有意水準 α を調整
- ステージ1の被験者内分散(だけ)を利用して、被験者数を再計算
- EMAのBEガイドラインでの言及、FDAの審査実績あり
- 簡便だが、**第一種の過誤確率を完全には制御不能**

Standard Combination Test (I)

逆正規法によるステージ1と2の統計量統合

- 理論的に第一種の過誤確率を制御
- ステージ1とステージ2の統計量を重み w ($0 < w < 1$)を用いて、逆正規法により統合した統計量を最終解析に利用

$$Z_{0j} = \sqrt{w}Z_{1j} + \sqrt{1-w}Z_{2j}$$

ここで、 $Z_{ij} = \Phi^{-1}(1 - p_{ij})$ とする

($\Phi^{-1}(\cdot)$ は標準正規分布の累積分布関数の逆関数)

Z_{1j} と Z_{2j} は独立 (ステージ1とステージ2は異なる被験者であるため)

また、重み w は試験計画時に設定

Standard Combination Test (2)

棄却限界値

- 最終解析の検定統計量 Z_{0j} および中間解析の検定統計量 Z_{1j} は、ポイント帰無仮説 H_{0j}^* のもとで、
正の相関 \sqrt{w} をもつ二変量標準正規分布に従う

$$\begin{pmatrix} Z_{1j} \\ Z_{0j} \end{pmatrix} \sim N \left(\begin{pmatrix} 0 \\ 0 \end{pmatrix}, \begin{pmatrix} 1 & \sqrt{w} \\ \sqrt{w} & 1 \end{pmatrix} \right)$$

- ステージ i 終了時の検定に対する棄却限界値を z_{α_i} として
試験全体での第一種の過誤確率が α となるように
 z_{α_i} を決定

$$P(Z_{1j} < z_{\alpha_1} \cap Z_{0j} < z_{\alpha_2}) = 1 - \alpha$$

Standard Combination Test (3)

BE判定方法

- 中間解析：

H_{01} および H_{02} が棄却されるとき

つまり, $p_{11} < \alpha_1$ かつ $p_{12} < \alpha_1$ のとき

BEと判定

- 最終解析：

逆正規法により統合した統計量 Z_{0j} に基づきBE評価

Z_{0j} の観測値 z_{0j} が $z_{01} \geq z_{\alpha_2}$ かつ $z_{02} \geq z_{\alpha_2}$ のとき

BEと判定

Maximum Combination Test (I)

Standard Combination Testの改良法

- 理論的に第一種の過誤確率を制御
- Standard Combination Testでは
重み $w = n_1 / (n_1 + n_2)$ のとき，検出力が最大に
- ステージ2の被験者数 n_2 はステージ1の結果に依存するため
事前に重み w の最適値は不明

Maximum Combination Test (2)

検定統計量

- 重みを w と w^* ($0 < w^* < w < 1$)の2つ設定し,
それぞれの重みに対する統計量 Z_{0j} と Z_{0j}^* を算出

$$Z_{0j} = \sqrt{w}Z_{1j} + \sqrt{1-w}Z_{2j}$$

$$Z_{0j}^* = \sqrt{w^*}Z_{1j} + \sqrt{1-w^*}Z_{2j}$$

- 最終解析では, $Z_{maxj} = \max(Z_{0j}, Z_{0j}^*)$ に基づき
BE評価

Maximum Combination Test (3)

棄却限界値

- 中間解析の検定統計量 Z_{1j} および最終解析の検定統計量 Z_{0j} , Z_{0j}^* はポイント帰無仮説 H_{0j}^* のもとで、以下の三変量標準正規分布に従う

$$\begin{pmatrix} Z_{1j} \\ Z_{0j} \\ Z_{0j}^* \end{pmatrix} \sim N \left(\begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} 1 & \sqrt{w} & \sqrt{w^*} \\ \sqrt{w} & 1 & \left(\sqrt{ww^*} + \sqrt{(1-w)(1-w^*)} \right) \\ \sqrt{w^*} & \left(\sqrt{ww^*} + \sqrt{(1-w)(1-w^*)} \right) & 1 \end{pmatrix} \right)$$

- ステージ1とステージ2終了時の検定に対する棄却限界値をそれぞれ $z_{\alpha_{1,max}}$ と z_{max} として、試験全体での第一種の過誤確率が α となるように $z_{\alpha_{1,max}}$ と z_{max} を決定

$$P \left(Z_{1j} < z_{\alpha_{1,max}} \cap Z_{0j} < z_{max} \cap Z_{0j}^* < z_{max} \right) = 1 - \alpha$$

Maximum Combination Test (4)

BE判定方法

- 中間解析：

H_{01} および H_{02} が棄却されるとき

つまり, $p_{11} < \alpha_{1,max}$ かつ $p_{12} < \alpha_{1,max}$ のとき

BEと判定

- 最終解析：

$Z_{max_j} = \max(Z_{0j}, Z_{0j}^*)$ に基づき BE評価

Z_{max_j} の観測値 z_{max_j} が $z_{max_1} \geq z_{max}$ かつ $z_{max_2} \geq z_{max}$ のとき

BEと判定

被験者数再計算アルゴリズム (1)

Potvinらの算出法とその問題点

- 試験計画時の被験者数計算の設定値から
ステージ1の被験者内分散のみを置き換えて
ステージ2の被験者数を計算^[8]
- Maurerらは、ステージ1の片側検定の p 値
(つまり、製剤間の平均値の差)を無視することは
ステージ2の被験者数を必要以上に増大にさせると指摘^[10]

被験者数再計算アルゴリズム (2)

Maurerらの提案法

- 下式を満たす最小のステージ2の被験者数 n_2 を求める

$$P \left(T_{21} \geq t_{n_2-2, \alpha_{max_1}^c} \cap T_{22} \geq t_{n_2-2, \alpha_{max_2}^c} \right) \geq 1 - \beta^c$$

$t_{\nu, \alpha}$: 自由度 ν である t 分布の上側累積確率 α 点

$\alpha_{max_j}^c$: conditional error rate for Maximum Combination Test

$1 - \beta^c$: estimated conditional target power

被験者数再計算アルゴリズム (3)

提案法のポイント：ステージIのp値を考慮

- conditional error rate (Maximum Combination Test)

ポイント帰無仮説 H_{0j}^* のもとで、ステージIのデータを所与としたとき
ステージ2で帰無仮説 H_{0j} を棄却する確率

$$\alpha_{max_j}^c = 1 - \Phi(\min_j)$$

$$\min_j = \min \left[\left(z_{max} - \sqrt{w} z_{1j} \right) / \sqrt{1 - w}, \left(z_{max} - \sqrt{w^*} z_{1j} \right) / \sqrt{1 - w^*} \right]$$

z_{max} : Maximum Combination Testの最終解析の棄却限界値

z_{1j} : ステージIの統計量 Z_{1j} の観測値

- estimated conditional target power

ステージIでBEと判定されなかったもとでステージ2でBEと判定される条件付き確率

$$1 - \beta^c = (\hat{\beta}_1 - \beta) / \hat{\beta}_1$$

$1 - \beta$: 試験計画時の検出力

$1 - \hat{\beta}_1$: ステージIの検出力の推定値

被験者数再計算アルゴリズム (4)

2変量非心 t 分布に基づく検出力の計算

- T_{21} と T_{22} は、相関係数が -1 、自由度 $n_2 - 2$ 、非心度がそれぞれ $(\log(1.25) + \delta_{ap})/\sqrt{2\hat{\sigma}_1^2/n_2}$ 、 $(\log(1.25) - \delta_{ap})/\sqrt{2\hat{\sigma}_1^2/n_2}$ である
2変量非心 t 分布に従うものとして検出力を計算

$$\delta_{ap} = \begin{cases} +|\delta_p| & \text{if } \alpha_{max_1}^c > \alpha_{max_2}^c \\ -|\delta_p| & \text{if } \alpha_{max_1}^c \leq \alpha_{max_2}^c \end{cases}$$

$$\delta_p = \log(\text{計画時の幾何平均比})$$

$\delta_p \neq 0$ であるとき、 $\delta' \in [-|\delta_p|, +|\delta_p|]$ である全ての δ' に対して

検出力が estimated conditional target power 以上となることを保証するための処理

Maurer らは [Adaptive Planning](#) と呼称

各解析手法の性能比較

Potvinの方法 vs Std. vs Max. Combination Test

- Maurerらはシミュレーションによる比較結果に基づき Maximum Combination Testを推奨^[10]

	第一種の過誤確率	検出力(80%目標)	期待被験者数
Potvinの方法 (Method B)	5%を越える場合あり (最大5.68%)	ステージ1のNが少ない場合や 被験者内変動係数が50%超の 場合は80%未滿	ステージ1の群間差を考慮しな いため, Std./Max. Comb. Testより少ない
Standard Combination Test	全ての条件で5%以下	ステージ1のNが 12名より多い場合は 80%以上	ステージ2の被験者数に上限を 設けない場合, 被験者内変動係 数が大きくなると被験者数が一 番多くなる
Maximum Combination Test	全ての条件で5%以下	ステージ1のNが 12名より多い場合は 概ね80%以上で, Std. Comb. Testに優る場合もある	想定からの乖離に対して Std. Comb. Testより堅牢



発表内容

1. 背景
2. BEガイドラインの改正に対応しうる解析手法
3. SASによる実装方法の提案
4. その他，考慮すべき事項

SASによる実装方法の提案（I）


本発表で作成したSASプログラム一覧

No.	プログラム名	内容
1	max_comb.sas	Maximum Combination Testの実装例
2	max_comb_simulation.sas	Maximum Combination Testの性能評価のシミュレーションの実装例
3	maurer_example.sas	MaurerらのAppendixのRプログラム ^[16] をSASに移植したもの
4	std_comb.sas	Standard Combination Testの実装例
5	potvin_example1.sas	Potvin ^[8] のExample1 (Method BおよびC)をSASで再現したもの
6	potvin_example2.sas	Potvin ^[8] のExample2 (Method BおよびC)をSASで再現したもの

※上記1～4は、検出力計算のためにDr.Frank Bretz作成のprobmvt.sasをinclude
また、プログラムの動作確認は、SAS OnDemand for Academicsにて実施

SASによる実装方法の提案（2）

想定する判定フロー^[10]



SASによる実装方法の提案（3）

想定する解析データセット構造

各被験者につき，時期1と時期2の情報を1レコードに格納

STAGE	USUBJID	TRTSEQA	TRTA1	TRTA2	lnCmax1	lnCmax2
1	xx-01	RT	R	T	X.XXXX	X.XXXX
1	xx-02	RT	R	T	X.XXXX	X.XXXX
:	:	:	:	:	:	:
1	xx-19	TR	T	R	X.XXXX	X.XXXX
1	xx-20	TR	T	R	X.XXXX	X.XXXX
2	xx-21	RT	R	T	X.XXXX	X.XXXX
:	:	:	:	:	:	:
2	xx-55	TR	T	R	X.XXXX	X.XXXX
2	xx-56	TR	T	R	X.XXXX	X.XXXX

SASによる実装方法の提案 (4)

Maximum Combination Testの棄却限界値の決定

- 本発表時点では、
三変量以上の標準正規分布の累積分布
に対応したSAS関数がない
- 三変量標準正規分布に従う正規乱数を
大量に発生させ、3つの検定統計量に
対する棄却限界値が等しいという制約
下で試験全体の第一種の過誤確率が
5%になるような棄却限界値および有
意水準を探索
- 計算精度を担保するために可能な限り
多くの乱数を発生させることが重要

```
proc iml;
  call randseed(12345678);
  randN = 100000000; /* recommend >= 100,000,000 */
  * choice of weights;
  w1 = 0.5; w2 = 1 - w1; w1_star = 0.25; w2_star = 1 - w1_star;
  root_w1 = sqrt(w1); root_w1_star = sqrt(w1_star);
  rho = sqrt(w1 * w1_star) + sqrt(w2 * w2_star);
  /* correlation matrix */
  Sigma = (1 || root_w1 || root_w1_star) //
           (root_w1 || 1 || rho) //
           (root_w1_star || rho || 1);
  mean = {0 0 0};
  Z = randnormal(randN, mean, Sigma); /* sample from MVN(0, Sigma) */
  create MVNorm3 from Z[c={Z1 Z0a Z0b}]; append from Z; close;

  start TVNCDF(x);
    use MVNorm3;
    read all into TRIVN;
    ind = (TRIVN[,1] < x) & (TRIVN[,2] < x) & (TRIVN[,3] < x);
    prob = ind[,];
    f = (prob) - (1 - 0.05);
    return (f);
  finish TVNCDF;
  * get critical value of robust combination test for z_max1 and z_max2;
  z_crit_max = froot("TVNCDF", {0, 5});
  * convert to nominal significance level;
  alpha_max = 1 - probnorm(z_crit_max);
  print z_crit_max alpha_max;
quit;
```

SASによる実装方法の提案 (5)

中間解析：2つの片側検定 (TOST)

- TTESTプロシジャの**TOSTオプション**を指定し、2つの片側検定のp値を出力
- 無効中止ルールに該当するか判定するための両側90%信頼区間および被験者数再設計を行うための各種統計量をODSステートメントで出力

```
ods output Statistics=stat1 ConfLimits=cl1 EquivLimits=ev1 EquivTests=et1;
proc ttest data=stage1plus2 tost(%sysfunc(log(0.80)), %sysfunc(log(1.25)))
alpha=0.05 order=internal;
  where STAGE = 1;
  var lnCmax1 lnCmax2 / crossover=(TRTA1 TRTA2) ;
run;
```

治療	期間	手法	分散	検定	Null	自由度	t 値	p 値
Diff (1-2)		Pooled	Equal	上限	-0.223	18	1.60	0.0632
Diff (1-2)		Pooled	Equal	下限	0.2231	18	-2.36	0.0150

SASによる実装方法の提案 (6)

条件付き検出力に基づく被験者数再計算

- 2変量非心 t 分布の確率計算は、Genz and Bretz^[17]のプログラム (probmvt.sas) を利用
- 被験者数を2例ずつ増やしながらステージ2の条件付き検出力が estimated conditional target power を達成する最小の被験者数を算出

```
%include "/home/u#####/probmvt.sas"; *specify the folder;
proc iml;
  load ; *load modules in probmvt.sas;
  lnGmrPlanned = log(0.95); lnlp25 = log(1.25);
  variancel = &var_hat_stagel.; *intra-subject variance on log-scale;
  COVAR = {1 -1,
            -1 1};
  acheived = 0;
  do N = %eval(&n_stagel.+2) to 200 by 2 until(acheived);
    DIM = 2; DF = N - 2;
    nc1 = (lnlp25 + lnGmrPlanned) /sqrt(variancel * 2 / N);
    nc2 = (lnlp25 - lnGmrPlanned) /sqrt(variancel * 2 / N);
    DELTA = j(1, 2); DELTA[1] = nc1; DELTA[2] = nc2;
    INFIN = J(1, DIM, 1);
    tc1 = tinv(1 - &alpha_c1_max., DF); *conditional error rate for H01;
    tc2 = tinv(1 - &alpha_c2_max., DF); *conditional error rate for H02;
    LOWER = j(1, 2); LOWER[1] = tc1; LOWER[2] = tc2;
    UPPER = j(1, 2, 999); *infinity:999 is ignored;
    MAXPTS = 2000*DIM*DIM*DIM; ABSEPS = .000001; RELEPS = 0;
    RUN MVN_DIST( DIM, DF, DELTA, LOWER, UPPER, INFIN, COVAR, MAXPTS,
                  ABSEPS, RELEPS, ERROR, condpower, NEVALS, INFORM );
    if (condpower >= &conditional_target_power.) then acheived = 1;
    print N condpower;
    if acheived = 1 then do;
      create SSR from N[c={N}]; append from N; close;
    end;
  end;
quit;
```


SASによる実装方法の提案 (7)

Maximum Combination TestによるBE判定 (最終解析)

- ステージ2のデータから
中間解析と同様にTTESTプロシジャで
各種統計量を算出
- 中間解析で求めたステージ1の統計量と
予め定めた2つの重みを用いて
逆正規法による統合を行い、
Maximum Combination Testによる
BE判定を実施

```
data maxcomb;
  merge tost1st tost2nd;
  w = 0.5;
  w_star = 0.25;
  z_crit_max = 1.9374;
  * z-statistics;
  z1_w = sqrt(w) * z1_stage1 + sqrt(1 - w) * z1_stage2;
  z2_w = sqrt(w) * z2_stage1 + sqrt(1 - w) * z2_stage2;
  z1_w_star = sqrt(w_star) * z1_stage1 +
    sqrt(1 - w_star) * z1_stage2;
  z2_w_star = sqrt(w_star) * z2_stage1 +
    sqrt(1 - w_star) * z2_stage2;
  z1_max = max(z1_w, z1_w_star);
  z2_max = max(z2_w, z2_w_star);

  * reject null hypothesis at left side?;
  nrejlw = (z1_max >= z_crit_max);

  * reject null hypothesis at right side?;
  nrejup = (z2_max >= z_crit_max);

  * reject null hypothesis at both sides?;
  * i.e., is ABE accepted?;
  z_passed = (nrejlw & nrejup); *1:accepted, 0:rejected;
run;
```

SASによる実装方法の提案（8）

Simulationによる性能評価

試験性能の把握にSimulationが有用

各種条件の比較に以下のような出力を推奨

設定	幾何 平均比	被験者内 変動係数 (%)	ステージ1で BEと判定す る確率	ステージ1で BEでないと 判定する確率	ステージ2へ 移行する確率	ステージ2で BEと判定す る確率	ステージ1ま たは2で BEと判定す る確率	期待 総被験者数
$n_1=24$ $w=0.5$ $w^*=0.25$	0.8	20	0.#####	0.#####	0.#####	0.#####	0.#####	##.##
		30	0.#####	0.#####	0.#####	0.#####	0.#####	##.##
		40	0.#####	0.#####	0.#####	0.#####	0.#####	##.##
	1.0	20	0.#####	0.#####	0.#####	0.#####	0.#####	##.##
		30	0.#####	0.#####	0.#####	0.#####	0.#####	##.##
		40	0.#####	0.#####	0.#####	0.#####	0.#####	##.##




発表内容

1. 背景
2. BEガイドラインの改正に対応しうる解析手法
3. SASによる実装方法の提案
4. その他，考慮すべき事項

その他, 考慮すべき事項 (1)

固定デザインか2段階デザインか

- Kazaらは判定フローを提案^[18]



その他、考慮すべき事項（2）

綿密な試験計画の重要性

- BEガイドラインは、試験計画への明記を要求
 - 中間解析手順／第一種の過誤確率の制御方法／
追加する例数の算出方法／追加データの併合手順
- 試験計画時に決定するパラメータ
 - Maximum Combination Testの2つの重み
 - ステージIの被験者数
 - 無効中止基準…等
- パラメータ検討のため
シミュレーションによる試験の動作性能の評価も必要

その他、考慮すべき事項（3）

実施医療機関等の選定への影響

- 予定被験者数が不確定
 - 中間解析で追加被験者数を決定または早期中止となる可能性
- 実施医療機関，CRO(開発業務受託機関)，
薬物濃度測定機関等の関連機関にも予め説明が必要
- 現実には追加可能な被験者数には上限があり，
被験者数の上限が，コストやスケジュール，
実施医療機関等での受託可否などに影響

その他, 考慮すべき事項 (4)

報告すべき事項

- 治験総括報告書等で詳細な報告が必要
- FDAガイダンス^[19]やACEチェックリスト^[20]が指針に

Adaptive Designs for Clinical Trials of Drugs and Biologics Guidance for Industry

U.S. Department of Health and Human Services
Food and Drug Administration
Center for Drug Evaluation and Research (CDER)
Center for Biologics Evaluation and Research (CBER)

November 2019
Biostatistics

Trials

METHODLOGY Open Access

The adaptive designs CONSORT extension (ACE) statement: a checklist with explanation and elaboration guideline for reporting randomised trials that use an adaptive design

Munyaziro Dimairo^{1*}, Philip Pallmann², James Wason^{3,4}, Susan Todd⁵, Thomas Jakl⁶, Steven A. Julious¹, Adrian P. Mander^{2,3}, Christopher J. Weir², Franz Koenig⁷, Marc K. Walton⁸, Jon P. Nicholl¹, Elizabeth Coates¹, Katie Biggs¹, Toshimitsu Hamaseki¹⁰, Michael A. Proschian¹¹, John A. Scott¹², Yuki Ando¹³, Daniel Hind¹, Douglas G. Altman¹⁴ and on behalf of the ACE Consensus Group

Abstract

Adaptive designs (ADs) allow pre-planned changes to an ongoing trial without compromising the validity of conclusions and it is essential to distinguish pre-planned from unplanned changes that may also occur. The reporting of ADs in randomised trials is inconsistent and needs improving. Incompletely reported AD randomised trials are difficult to reproduce and are hard to interpret and synthesise. This consequently hampers their ability to inform practice as well as future research and contributes to research waste. Better transparency and adequate reporting will enable the potential benefits of ADs to be realised.

This extension to the Consolidated Standards Of Reporting Trials (CONSORT) 2010 statement was developed to enhance the reporting of randomised AD clinical trials. We developed an Adaptive designs CONSORT Extension (ACE) guideline through a two-stage Delphi process with input from multidisciplinary key stakeholders in clinical trials research in the public and private sectors from 21 countries, followed by a consensus meeting. Members of the CONSORT Group were involved during the development process.

The paper presents the ACE checklists for AD randomised trial reports and abstracts, as well as an explanation with examples to aid the application of the guideline. The ACE checklist comprises seven new items, nine modified items, six unchanged items for which additional explanatory text clarifies further considerations for ADs, and 20 unchanged items not requiring further explanatory text. The ACE abstract checklist has one new item, one modified item, one unchanged item with additional explanatory text for ADs, and 15 unchanged items not requiring further explanatory text.

The intention is to enhance transparency and improve reporting of AD randomised trials to improve the

(Continued on next page)

Table 2 ACE checklist for the main report

Section/Topic	Item No	Standard CONSORT 2010 checklist item	Extension for adaptive design randomised trials	Page No
Title and abstract	1a	Identification as a randomised trial in the title		
	1b	Structured summary of trial design, methods, results, and conclusions (for specific guidance see CONSORT for abstracts) [136, 137]	Structured summary of trial design, methods, results, and conclusions (for specific guidance see ACE for abstracts, Table 3)	
Introduction				
	2a	Scientific background and explanation of rationale		
Methods	2b	Specific objectives or hypotheses		
	3a	Description of trial design (such as parallel, factorial) including allocation ratio		
Participants	3b ^a		Type of adaptive design used, with details of the pre-planned trial adaptations and the statistical information informing the adaptations	
	3c3b ^b	Important changes to methods after trial commencement (such as eligibility criteria), with reasons	Important changes to the design or methods after trial commencement (such as eligibility criteria) outside the scope of the pre-planned adaptive design features, with reasons	
Interventions	4a	Eligibility criteria for participants		
	4b	Settings and locations where the data were collected		
Outcomes	5	The interventions for each group with sufficient details to allow replication, including how and when they were actually administered		
	6a ^b	Completely defined pre-specified primary and secondary outcome measures, including how and when they were assessed	Completely define pre-specified primary and secondary outcome measures, including how and when they were assessed. Any other outcome measures used to inform pre-planned adaptations should be described with the rationale	
Sample size and operating characteristics	6b ^b	Any changes to trial outcomes after the trial commenced, with reasons	Any unplanned changes to trial outcomes after the trial commenced, with reasons	
	7a ^b	How sample size was determined	How sample size and operating characteristics were determined	
	7b ^c	When applicable, explanation of any interim analyses and stopping guidelines	Pre-planned interim decision-making criteria to guide the trial adaptation process; whether decision-making criteria were binding or non-binding; pre-planned and actual timing and frequency of interim data looks to inform trial adaptations	



まとめ

令和2年3月の後発医薬品のBEガイドラインの改正により認められた被験者数再計算を伴う2段階デザインに対応した解析手法として Maximum Combination Testおよび被験者数再計算アルゴリズムを関連手法とともに紹介し，SASによる実装方法を提案した

本発表が，SASユーザーの皆様の一助になれば幸いです

引用文献


1. 後発医薬品の生物学的同等性試験ガイドライン(薬生薬審発0319第1号, 別紙1)
2. 後発医薬品の生物学的同等性試験ガイドラインQ&A(令和2年3月19日事務連絡, 別紙1)
3. 後発医薬品 品質情報 No.14 令和2年8月(厚生労働省 医薬・生活衛生局医薬品審査管理課)
4. 含量が異なる経口固形製剤の生物学的同等性試験ガイドライン(薬生薬審発0319第1号, 別紙2)
5. 経口固形製剤の処方変更の生物学的同等性試験ガイドライン(薬生薬審発0319第1号, 別紙3)
6. 剤形が異なる製剤の追加のための生物学的同等性試験ガイドライン(薬生薬審発0319第1号, 別紙4)
7. Kuribayashi R, Yamaguchi T, Takagi K. Modernization and Strengthening of Bioequivalence Guidelines in Japan. Clin Pharmacokinet. 2021 Feb;60(2):145-151.
8. Potvin D, DiLiberti CE, Hauck WW, Parr AF, Schuirmann DJ, Smith RA. Sequential design approaches for bioequivalence studies with crossover designs. Pharm Stat. 2008;7(4):245-62.
9. Montague TH, Potvin D, Diliberti CE, Hauck WW, Parr AF, Schuirmann DJ. Additional results for Sequential design approaches for bioequivalence studies with crossover designs. Pharm Stat. 2012;11(1):8-13.
10. Maurer W, Jones B, Chen Y. Controlling the type I error rate in two-stage sequential adaptive designs when testing for average bioequivalence. Stat Med. 2018;37(10):1587-607.
11. 棚橋 昌也, 菅波 秀規. 中間解析, サンプルサイズ再計算を伴う生物学的同等性試験における統計的課題. ジェネリック研究2021 ;15: 020-030.
12. Xu J, Audet C, Diliberti CE, Hauck WW, Montague TH, Parr AF, et al. Optimal adaptive sequential designs for crossover bioequivalence studies. Pharm Stat. 2016;15(1):15-27.
13. European Medicine Agency. Committee for Medicinal Products for Human Use (CHMP) Guideline on the investigation of bioequivalence. CPMP/EWP/QWP/1401/98 Rev 1/ Corr **. 2010 January 20 .
14. Lee J, Feng K, Xu M, Gong X, Sun W, Kim J, Zhang Z, Wang M, Fang L, Zhao L. Applications of Adaptive Designs in Generic Drug Development. Clin Pharmacol Ther. 2021 Jul;110(1):32-35.
15. Kieser M, Rauch G. Two-stage designs for cross-over bioequivalence trials. Stat Med. 2015;34(16):2403-2416.
16. Maurer W, Jones B, Chen Y. Supplementary Information for "Controlling the Type I error rate in two-stage sequential adaptive designs when testing for Average Bioequivalence"
<https://onlinelibrary.wiley.com/action/downloadSupplement?doi=10.1002%2Fsim.7614&file=sim7614-sup-0001-supplementary.pdf>
17. Genz, A. and Bretz, F. Methods for the computation of multivariate t-probabilities. Journal of Computational and Graphical Statistics, 2002; 11, 950-971.
18. Kaza M, Sokolovskiy A, Rudzki PJ. 10th Anniversary of a Two-Stage Design in Bioequivalence. Why Has it Still Not Been Implemented? Pharm Res. 2020 Jul 13;37(7):140.
19. Food and Drug Administration. Adaptive designs for clinical trials of drugs and biologics: guidance for industry. 2019.
(邦訳: https://www.jpma.or.jp/information/evaluation/results/allotment/lofurc000000n5a8-att/adaptive_design.pdf)
20. Dimairo M, Pallmann P, Wason J, Todd S, Jaki T, Julious SA, Mander AP, Weir CJ, Koenig F, Walton MK, Nicholl JP, Coates E, Biggs K, Hamasaki T, Proschan MA, Scott JA, Ando Y, Hind D, Altman DG; ACE Consensus Group. The Adaptive designs CONSORT Extension (ACE) statement: a checklist with explanation and elaboration guideline for reporting randomised trials that use an adaptive design. BMJ. 2020 Jun 17;369:m115.



Back Up

- Potvinの方法 (Method C/D)
- Adaptive Planningが必要な理由
- Standard Combination Testの棄却限界値の決定




Potvinの方法 (Method C/D)



被験者数再計算アルゴリズム（おまけ）

Adaptive Planningが必要な理由

Conditional Error Rateが不等($\alpha_{max_1}^c \neq \alpha_{max_2}^c$)の場合，検出力曲線が左右非対称となり
 $+|\delta_p|$ と $-|\delta_p|$ で検出力が異なる．したがって， Adaptive Planningによって
 検出力が低い方が目標検出力を越えることを保証



SASによる実装方法の提案（おまけ）

Standard Combination Testの棄却限界値の決定

Standard Combination Testの検定統計量は、ポイント帰無仮説 H_{0j}^* のもとで二変量標準正規分布に従うため、棄却限界値の決定にはPROBBNRM関数が利用可能

```
proc iml;
    * function to solve for critical value (equal alphas in stages 1 and 2);
    start BVNCDF(x);
        w1_star = 0.25;
        f = PROBBNRM(x, x, sqrt(w1_star)) - (1 - 0.05);
        return (f);
    finish BVNCDF;

    * get critical value of standard combination test;
    z_crit_stagel_star = froot("BVNCDF", {0, 5});
    * convert this to a significance level;
    alpha_stagel_star = 1 - probnorm(z_crit_stagel_star);

    print z_crit_stagel_star alpha_stagel_star;
quit;
```